clear

*Load Data
use "./data_geo.dta", clear

*****************************************************************************
*** Setup: see Kashyap & Stein (2000) and/or Den Haan, Sumner, Yamashiro (2002) for details ***

*Drop non-states 
drop if rssd9200 == "0"
keep if  rssd9210>0 & rssd9210< 57

*Drop Delaware and/or South Dakota (see Jayaratne & Strahan 1998 JLE)
*drop if rssd9210 == 10
*drop if rssd9210 == 46

*Sample period
keep if year_q < tq(1995q1)

*Positive Assets
keep if rcfd2170 > 0

*Commercial Banks Only
keep if rssd9048 == 200 

*Insured
*keep if rssd9424 == 1 | rssd9424 == 2 | rssd9424 == 6
drop if rssd9424 == 0

*Merger
drop if MERGE_CD != .

*Set panel data
xtset rssd9001 year_q

*Fix Total Loans pre-1984 (insert before calcuating loans2 above)
gen loans2 = rcfd1400
replace loans2 = rcfd1400+rcfd2165 if year_q < tq(1984q1)

*Put loans and loan components in real terms
gen c_p_i = cpi/100
gen r_loans = loans2/c_p_i 
gen r_ci = rcfd1600/c_p_i
gen r_rlest = rcfd1410/c_p_i
gen r_con = rcfd1975/c_p_i
gen r_ag = rcfd1590/c_p_i

*Log difference 
gen loans = ln(r_loans) - ln(L.r_loans)
gen ci = ln(r_ci) - ln(L.r_ci) 
gen re = ln(r_rlest) - ln(L.r_rlest) 
gen con = ln(r_con) - ln(L.r_con)
gen ag = ln(r_ag) - ln(L.r_ag) 

*Drop outliers
sort year_q
by year_q: egen sdv = sd(loans)
drop if loans >= 5*sdv
drop if loans <= -5*sdv

*Loan Component Outliers
gen ci_share = rcfd1600/loans2
*drop if ci_share < 0.05
gen re_share = rcfd1410/loans2
*drop if re_share < 0.05
gen con_share = rcfd1975/loans2
*drop if con_share < 0.05
gen ag_share = rcfd1590/loans2
*drop if ag_share < 0.05

*Four Consecutive Quarters of Loan Growth
*ssc install tsspell
tsspell, c(loans > 0)
sort rssd9001
by rssd9001: egen maxrun = max(_seq)
drop if maxrun < 4

*Percentiles
sort year_q
by year_q: egen p95 = pctile(rcfd2170), p(95)
by year_q: egen p99 = pctile(rcfd2170), p(99)

*Generate Quarter/Year/timetrend Series
sort rssd9001 year_q
gen Quarter = quarter(date2)
gen Year = year(date2) 
egen time = group(year_q)

*Choose to keep (or drop) year deregulation was implemented 
replace intra = 1 if intra == .
replace inter =1 if inter == .

* Generate Interaction Terms
gen rr_intra = rr*intra
gen rr_intra1 = L1.rr*intra
gen rr_intra2 = L2.rr*intra
gen rr_intra3 = L3.rr*intra
gen rr_intra4 = L4.rr*intra

gen rr_inter = rr*inter
gen rr_inter1 = L1.rr*inter
gen rr_inter2 = L2.rr*inter
gen rr_inter3 = L3.rr*inter
gen rr_inter4 = L4.rr*inter

******************************************************************************
*** Table A.4 ***

gen gm_dum = 0
replace gm_dum = 1 if year_q > tq(1983q4)
gen rr_gm = rr*gm_dum
gen rr_gm1 = L1.rr*gm_dum
gen rr_gm2 = L2.rr*gm_dum
gen rr_gm3 = L3.rr*gm_dum
gen rr_gm4 = L4.rr*gm_dum

xtreg loans L(1/4).loans L(0/4).gdp_chg L(0/4).pce_chg L(0/4).pi_pchg L(0/4).hpi_chg L(0/4).vwretx L(0/4).rr gm_dum rr_gm rr_gm1 rr_gm2 rr_gm3 rr_gm4 i.Quarter i.Year, fe vce(cluster rssd9210) nonest
lincom rr + L1.rr + L2.rr + L3.rr + L4.rr
lincom rr_gm + rr_gm1 + rr_gm2 + rr_gm3 + rr_gm4

xtreg loans L(1/4).loans L(0/4).gdp_chg L(0/4).pce_chg L(0/4).pi_pchg L(0/4).hpi_chg L(0/4).vwretx L(0/4).rr gm_dum inter rr_gm rr_gm1 rr_gm2 rr_gm3 rr_gm4 rr_inter rr_inter1 rr_inter2 rr_inter3 rr_inter4 i.Quarter i.Year, fe vce(cluster rssd9210) nonest
lincom rr + L1.rr + L2.rr + L3.rr + L4.rr
lincom rr_gm + rr_gm1 + rr_gm2 + rr_gm3 + rr_gm4
lincom rr_inter + rr_inter1 + rr_inter2 + rr_inter3 + rr_inter4


